Als concreet voorbeeld laten we Maple8 opgave 2002-3 -C van de **UWC** oplossen. [De gepubliceerde oplossing met behulp van inductie, van *Hubrechts* en *Delvaux*, staat op p.93, deel 4, nr.1 (maart 2003)]. De identiteit luidt:

$$\sum_{j=0}^{k} \frac{\mathrm{binomial}(2\,j, j)\,\mathrm{binomial}(n-2\,j, k-j)}{j+1} = \mathrm{binomial}(n+1, k)$$

```
> restart:assume(n,integer,k,integer):assume(k,nonnegative,n>k):
  with(SumTools[Hypergeometric]);
```

[*AccurateSummation*, *AreSimilar*, *CanonicalRepresentation*, *DefiniteSum*, *ExtendedGosper*, *Gosper*, *IndefiniteSum*, *IsHolonomic*, *IsHypergeometricTerm*, *IsProperHypergeometricTerm*, *IsZApplicable*, *MultiplicativeDecomposition*, *PolynomialNormalForm*, *RationalCanonicalForm*, *SumDecomposition*, *Verify*, *WZMethod*, *Zeilberger*, *ZeilbergerRecurrence*, *ZpairDirect*]

Door met de cursor op een van bovenstaande termen te gaan staan en het helpmenu in de taakbalk van Maple aan te klikken kunt u direct gedetailleerde informatie over deze term verkrijgen. We laten Maple nu de som exact uitrekenen.

```
> TT:=binomial(2*j,j)*binomial(n-2*j,k-j)/(j+1);
  IsProperHypergeometricTerm(TT,n,k);
```

$$TT := \frac{\mathrm{binomial}(2\,j, j)\,\mathrm{binomial}(n\sim-2\,j, k\sim-j)}{j+1}$$

$$\mathit{true}$$

```
> Sum(TT,j=0..k)=simplify(DefiniteSum(TT,k,j,0..k));
```

$$\sum_{j=0}^{k\sim} \frac{\mathrm{binomial}(2\,j, j)\,\mathrm{binomial}(n\sim-2\,j, k\sim-j)}{j+1} = \frac{(-1)^{k\sim}\, n\sim \sin(\pi\, n\sim)\,(n\sim^2-1)\,\Gamma(n\sim-1)\,\Gamma(-n\sim-1+k\sim)}{\pi\,\Gamma(k\sim+1)}$$

```
> simplify(%,GAMMA);
```

$$\sum_{j=0}^{k\sim} \frac{\Gamma(2\,j+1)\,\Gamma(n\sim+1-2\,j)}{\Gamma(j+1)^2\,\Gamma(k\sim-j+1)\,\Gamma(n\sim+1-j-k\sim)\,(j+1)} = \frac{(-1)^{k\sim}\sin(\pi\, n\sim)\,\Gamma(n\sim+2)\,\Gamma(-n\sim-1+k\sim)}{\pi\,\Gamma(k\sim+1)}$$

Het rechterlid moet natuurlijk in de zin van een limiet geïnterpreteerd worden rekening houdend met de spiegelingsformule voor de Gammafunctie

```
> 'GAMMA(z)*GAMMA(1-z)'=simplify(GAMMA(z)*GAMMA(1-z));
```

$$\Gamma(z)\,\Gamma(1-z) = \frac{\pi}{\sin(\pi\, z)}$$

Maar we willen liever Maple al het werk laten doen en informeren deze keer tegelijk naar de gebruikte methode, die, in principe, ook een bewijs levert.

```
> T:=(n+1-k)!*binomial(2*j,j)*binomial(n-2*j,k-j)/(j+1);
  infolevel[DefiniteSum]:=3:
```

```
Sum(T,j=0..k)=simplify(DefiniteSum(T,k,j,0..k));
```

$$T := \frac{(n\sim - k\sim + 1)!\,\mathrm{binomial}(2\,j, j)\,\mathrm{binomial}(n\sim - 2\,j, k\sim - j)}{j+1}$$

```
DefiniteSum:   "try algorithms for definite sum"
Definite:   "Construct the Zeilberger's recurrence"
Definite:   "Solve the recurrence equation ..."
Definite:   "Find a particular d'Alembertian solution"
Definite:   "Solve the homogeneous linear recurrence equation"
Definite:   "Construction of the general solution successful"
Definite:   "Solve the initial-condition problem"
```

$$\sum_{j=0}^{k\sim} \frac{(n\sim - k\sim + 1)!\,\mathrm{binomial}(2\,j, j)\,\mathrm{binomial}(n\sim - 2\,j, k\sim - j)}{j+1} = \frac{\Gamma(n\sim + 2)}{\Gamma(k\sim + 1)}$$

Nu werkt het perfekt. We illustreren nog even enkele aspecten van de gebruikte algorithme. Zie ook de betreffende helpbestanden in Maple8 en de verwijzingen naar de literatuur daarin.

```
> CanonicalRepresentation[2](T,n,k,'listform');
```

$$\left[\frac{((-j+1)!)^2\,(1-2\,j)}{(-j+1)^2\,(1-2\,j)!}, \frac{(n\sim - 2\,j)!\,(n\sim - k\sim + 1)!}{(k\sim - j)!\,(n\sim - j - k\sim)!}\right]$$

```
> Zpair:=Zeilberger(T,n,k,nShift);
```

$$\begin{aligned}
Zpair := [\,& n\sim + 1 - 2\,j + (j - 4 - n\sim)\,nShift + nShift^2, -(\\
& (n\sim - k\sim + 1)!\,\mathrm{binomial}(n\sim - 2\,j, k\sim - j)\,n\sim + (n\sim - k\sim + 1)!\,\mathrm{binomial}(n\sim - 2\,j, k\sim - j)\\
& - 2\,(n\sim - k\sim + 1)!\,\mathrm{binomial}(n\sim - 2\,j, k\sim - j)\,j\\
& + (2 + n\sim - k\sim)!\,\mathrm{binomial}(n\sim + 1 - 2\,j, k\sim - j)\,j\\
& - 4\,(2 + n\sim - k\sim)!\,\mathrm{binomial}(n\sim + 1 - 2\,j, k\sim - j)\\
& - (2 + n\sim - k\sim)!\,\mathrm{binomial}(n\sim + 1 - 2\,j, k\sim - j)\,n\sim\\
& + (3 + n\sim - k\sim)!\,\mathrm{binomial}(n\sim + 2 - 2\,j, k\sim - j))\,\mathrm{binomial}(2\,j, j)\,(2 + n\sim - k\sim)\,(j - k\sim) / (\\
& (j+1)\,(n\sim j - k\sim j + j + n\sim - 3\,k\sim + k\sim^2 + 2 - n\sim k\sim))\,]
\end{aligned}$$

```
> ZeilbergerRecurrence(T,n,k,f,0..n+1);
```

$$\begin{aligned}
& (n\sim + 1 - 2\,j)\,\mathrm{f}(n\sim) + (j - 4 - n\sim)\,\mathrm{f}(n\sim + 1) + \mathrm{f}(n\sim + 2) = -\Gamma(n\sim - 2\,j)\,j\,(-n\sim + 2\,j)\\
& (-n\sim - 1 + 2\,j)\,(n\sim^3\,\Gamma(n\sim)\,j + 3\,j\,n\sim^2\,\Gamma(n\sim) + 2\,j\,n\sim\,\Gamma(n\sim) - n\sim^4\,\Gamma(n\sim) - 6\,\Gamma(n\sim)\,n\sim^3\\
& - 11\,\Gamma(n\sim)\,n\sim^2 + 1 - 6\,\Gamma(n\sim)\,n\sim)\,\Gamma(2\,j+1) \Big/ (\Gamma(-j+1)\,\Gamma(n\sim + 3 - j)\,\Gamma(j+1)^2\\
& (-n\sim - 3 + j))
\end{aligned}$$

Een recursie van de orde 2. Controle:

```
> Verify(T,Zpair,n,k,nShift);
```

$$true$$